

Calcolo differenziale

- Note

- Autore

Claudio Marsan
Liceo Cantonale di Mendrisio
Via Agostino Maspoli
CH-6850 Mendrisio (Switzerland)
e-mail: claudio.marsan@liceomendrisio.ch

- Versione

Versione 2.0, 16 marzo 2003
Maple V Release 6.02 for Windows 2000

```
> restart:
```

Definiamo una funzione:

```
> f := x -> 3*x^4 + 2*x^3 - x + 1;
```

$$f := x \rightarrow 3x^4 + 2x^3 - x + 1$$

Per calcolare la derivata di f si usa il comando "diff":

```
> diff(f(x), x);
```

$$12x^3 + 6x^2 - 1$$

Esiste anche il comando "Diff", che è la forma inerte di "diff":

```
> Diff(f(x), x) = diff(f(x), x);;
```

$$\frac{\partial}{\partial x} (3x^4 + 2x^3 - x + 1) = 12x^3 + 6x^2 - 1$$

Se vogliamo avere derivate di ordine superiore si appende al nome della variabile il simbolo"\$" seguito dall'ordine della derivata (senza spazi). La derivata seconda della funzione f:

```
> diff(f(x), x$2);
```

$$36x^2 + 12x$$

Attenzione: il risultato del comando "diff" non è una funzione, ma un'espressione che può essere trasformata in funzione tramite il comando "unapply":

```
> f1 := diff(f(x), x);
```

$$f1 := 12x^3 + 6x^2 - 1$$

```
> f1(2); # Sbagliato!
```

$$12x(2)^3 + 6x(2)^2 - 1$$

```
> f1 := unapply(f1, x);
```

$$f1 := x \rightarrow 12x^3 + 6x^2 - 1$$

```
> f1(2); # Adesso va bene!
```

Con "diff" si possono derivare anche delle espressioni:

```
> g := 3*sin(x^2) - x*exp(x);
```

$$g := 3 \sin(x^2) - x e^x$$

```
> diff(g, x);
```

$$6 \cos(x^2) x - e^x - x e^x$$

Si possono derivare le funzioni (solo le funzioni) anche tramite l'operatore "D":

```
> D(f);
```

$$x \rightarrow 12 x^3 + 6 x^2 - 1$$

Il risultato di "D" è una funzione:

```
> df := D(f);
```

$$df := x \rightarrow 12 x^3 + 6 x^2 - 1$$

```
> df(2);
```

119

Derivate di ordine superiore con l'operatore "D":

```
> (D@@2)(f);
```

$$x \rightarrow 36 x^2 + 12 x$$

La derivata seconda nel punto x=5:

```
> (D@@2)(f)(5);
```

960

```
> restart;
```

Maple conosce le regole di derivazione:

```
> D(f + g);
```

$$D(f) + D(g)$$

```
> D(f - g);
```

$$D(f) - D(g)$$

```
> D(f * g);
```

$$D(f) g + f D(g)$$

```
> D(f / g);
```

$$\frac{D(f)}{g} - \frac{f D(g)}{g^2}$$

```
> normal(%);
```

$$-\frac{-D(f) g + f D(g)}{g^2}$$

```
> restart;
```

Per risolvere in modo esatto le equazioni differenziali si usa il comando "dsolve":

```
> eqdiff := 3*diff(y(t), t) + y(t) - 2*t = 0;
```

$$eqdiff := 3 \left(\frac{\partial}{\partial t} y(t) \right) + y(t) - 2t = 0$$

```
> dsolve(eqdiff, y(t));
```

$$y(t) = 2t - 6 + e^{(-1/3t)} _C1$$

Se si assegna una condizione iniziale allora la sintassi di "dsolve" muta leggermente:

```
> initcond := y(0) = 0;
```

$$initcond := y(0) = 0$$

```
> dsolve({eqdiff, initcond}, y(t));
```

$$y(t) = 2t - 6 + 6e^{(-1/3t)}$$

Se abbiamo bisogno della funzione y(t) per ulteriori calcoli oppure per fare un grafico o per altro si può procedere come segue:

```
> sol := dsolve({eqdiff, initcond}, y(t));
```

$$sol := y(t) = 2t - 6 + 6e^{(-1/3t)}$$

```
> y1 := unapply(rhs(sol), t);
```

$$y1 := t \rightarrow 2t - 6 + 6e^{(-1/3t)}$$

```
> y1(3);
```

$$6e^{(-1)}$$

Nel seguito vediamo l'influenza della scelta della condizione iniziale sulla soluzione:

```
> sol2 := dsolve({eqdiff, y(0) = 2}, y(t));
```

$$sol2 := y(t) = 2t - 6 + 8e^{(-1/3t)}$$

```
> y2 := unapply(rhs(sol2), t);
```

$$y2 := t \rightarrow 2t - 6 + 8e^{(-1/3t)}$$

```
> sol3 := dsolve({eqdiff, y(0) = 4}, y(t));
```

$$sol3 := y(t) = 2t - 6 + 10e^{(-1/3t)}$$

```
> y3 := unapply(rhs(sol3), t);
```

$$y3 := t \rightarrow 2t - 6 + 10e^{(-1/3t)}$$

```
> plot([y1(x), y2(x), y3(x)], x=-5..8, y=-3..10, color=[red,  
green, blue]);
```

